Πανεπιστήμιο Ιωαννίνων
Τμήμα Μηχ. Η/Υ και Πληροφορικής
"Μεταφραστές"
Διδάσκων: Γ. Μανής
Φεβρουάριος 2017

Γραμματική της Ciscal

<PROGRAM> ::= **program** ID <BLOCK>

<BLOCK> ::= **{** <DECLARATIONS> <SUBPROGRAMS> <SEQUENCE> **}**

<DECLARATIONS> ::= ε | **declare** <VARLIST> **enddeclare**

<VARLIST> ::= ε | ID (, ID)*

<SUBPROGRAMS> ::= (<FUNC>) *

<FUNC> ::= **procedure** ID <FUNCBODY> |

function ID <FUNCBODY>

<FUNCBODY> ::= <FORMALPARS> <BLOCK>

<FORMALPARS> ::= **(** ε | <FORMALPARLIST> **)**

<FORMALPARLIST> ::= <FORMALPARITEM> (, <FORMALPARITEM>)*

<FORMALPARITEM> ::= **in** ID | **inout** ID

<SEQUENCE> ::= <STATEMENT> (; <STATEMENT>)*

<BRACKETS-SEQ> ::= { <SEQUENCE> }

<BRACK-OR-STAT> ::= <BRACKETS-SEQ> | <STATEMENT>

<STATEMENT> ::= ε |

<ASSIGNMENT-STAT> |

<IF-STAT> |

<DO-WHILE-STAT> |

<WHILE-STAT> |

<SELECT-STAT> |

<EXIT-STAT> |

<RETURN-STAT> |

<PRINT-STAT> |

<CALL-STAT>

<ASSIGNMENT-STAT> ::= ID **:=** <EXPRESSION>

<IF-STAT> ::= **if (<**CONDITION>**)** <BRACK-OR-STAT> <ELSEPART>

<ELSEPART> ::= ε | **else** <BRACK-OR-STAT>

<WHILE-STAT> ::= **while (<**CONDITION>**)** <BRACK-OR-STAT>

<SELECT-STAT> ::= **select (**ID**)**

(CONST**:** <BRACK-OR-STAT>)*

DEFAULT <BRACK-OR-STAT>

<DO-WHILE-STAT> ::= **do** <BRACK-OR-STAT> **while (<**CONDITION>**)**

<EXIT-STAT> ::= **exit**

<RETURN-STAT> ::= **return (**<EXPRESSION>**)**

<PRINT-STAT> ::= **print (**<EXPRESSION>**)**

<CALL-STAT> ::= call ID <ACTUALPARS>

<ACTUALPARS> ::= **(** ε | <ACTUALPARLIST> **)**

<ACTUALPARLIST> ::= <ACTUALPARITEM> (**,** <ACTUALPARITEM>)*

<ACTUALPARITEM> ::= **in** <EXPRESSION> | **inout** ID

<CONDITION> ::= <BOOLTERM> (**or** <BOOLTERM>)*

<BOOLTERM> ::= <BOOLFACTOR> (**and** <BOOLFACTOR>)*

<BOOLFACTOR> ::=**not [**<CONDITION>**]** |

[<CONDITION>**]** |

<EXPRESSION> <RELATIONAL-OPER> <EXPRESSION>

<EXPRESSION> ::= <OPTIONAL-SIGN> <TERM> (<ADD-OPER> <TERM>)*

<TERM> ::= <FACTOR> (<MUL-OPER> <FACTOR>)*

<FACTOR> ::= CONSTANT |

(<EXPRESSION>**)** |

ID <IDTAIL>

<IDTAIL> ::= ε | <ACTUALPARS>

<RELATIONAL-OPER> ::= **=** | **<** | **<=** | **<>** | **>=** | **>**

<ADD-OPER> ::= **+** | **-**

<MUL-OPER> ::= ***** | **/**

<OPTIONAL-SIGN> ::= ε | <ADD-OPER>